
Scope and validity
This manual describes the HOYAN Rule Language (HRL) used by the rule engine built into the HOYAN ETH-A2

communication gateway, covering its syntax, execution model and operating limits. It is intended for control

engineers and system integrators responsible for configuring on-device logic.

The content of this document applies to firmware version V2.77. The behaviour, capacity limits and diagnostic

interfaces described here refer to that version and are subject to change in later releases without notice. Where this

document and the actual behaviour of the device differ, the behaviour of the device governs.

HRL is a declarative rule language, not IEC 61131-3. There are no function blocks, no ladder diagrams, no TON /

TOF / R_TRIG / SR / CTU. Everything is built from three primitives:

FOR <t>  — the condition must hold for a duration

ON RISE  / ON FALL  — edge triggering

$VAR.0 – $VAR.63  — 64 int32 scratch variables, visible live within the same scan

The parser is line-by-line with no AST. Unrecognised lines are silently ignored — no error is raised. A mistyped

keyword therefore quietly removes half a rule's meaning. Always verify rule count and contents after import.

NOTICE

The rule engine in this product implements general-purpose control logic and is not a safety-rated (SIL / PL)

device. Functions concerning personnel safety, prevention of equipment damage or environmental protection

must be implemented in an independent hardware safety circuit and must not rely on the rule engine alone.

File structure and limits

# comment lines start with #
RULE <name up to 23 chars> :
  IF   <tag> <op> <literal>  [FOR <duration>]
  AND  <tag> <op> <literal>
  AND  <tag> <op> <literal>
  THEN <action>
  AND  <action>
  AND  <action>
  ON   LEVEL | RISE | FALL | CHANGE | BOOT | TRIGGER | OFFDELAY
  COOLDOWN <duration>
  GROUP <n>    PRIORITY <n>    DISABLED

Indentation carries no meaning. Keywords are case-insensitive. RULE  opens a new rule; no END  is needed. One

directive per line — the parser consumes the first directive on a line and jumps straight to the next line.

Whether AND  / OR  fills a condition slot or an action slot depends solely on whether THEN  has been seen yet. That is

the parser's only state transition.

Item Limit Notes

Rules 160

Rule name 23 bytes ASCII recommended

Tag name 23 bytes

Conditions per rule 3 Left-associative, no parentheses

Item Limit Notes

Actions per rule 3 Excess clauses silently dropped

Groups 32 Can be enabled/disabled as a unit

User variables $VAR.0 – $VAR.63 32-bit signed, retained across power cycles

Active alarms 16

Trigger history 64 entries the History button on the Rules page

FOR  / COOLDOWN  duration 65535 ms i.e. 65.5 s

Output points per scan 96 Excess counted in $SYS.WQ_FAIL

Rules per import 64 Import more in append chunks

Import request size 4096 bytes Headers included; about 3.5 KB of HRL text

TRAP

Clauses beyond the third condition or third action are 

SILENTLY DISCARDED

. A fourth AND  produces no error at all — the line simply does not exist. Split into several rules chained through 

$VAR  instead.

TRAP — IMPORT SIZE CEILING

A single import request, headers included, must not exceed 

4096 BYTES

. Anything beyond that is 

SILENTLY DISCARDED

and the device returns no error. After headers, roughly 

3.5 KB

of HRL text is usable per import.

For larger projects, strip comments and split on rule boundaries: send the first chunk with mode: "replace"  and

the rest with mode: "append" . Append upserts sequentially after the existing rules, so execution order is

preserved exactly.

RULE declaration
The name runs to the :  or end of line, trailing spaces are stripped, and it is truncated to 23 characters. The colon is

optional.

The name is the lookup key for the TRIGGER  action and /api/rules/fire  (exact match on the name). With duplicate

names only the first is ever found.

Defaults for a fresh rule: enable=1 , priority=50 , expr*=NONE , action3=NOOP .

Use ASCII for rule names. The field is 24 bytes; a UTF-8 CJK character costs 3 bytes, allowing only 7 characters, and

truncation happens on a byte boundary — which can produce an invalid UTF-8 sequence.

IF conditions

Basic conditions

IF DI16_#2.DI0 == 1                  # digital input closed
IF DI16_#2.DI0 == ON                 # identical, ON parses to 1
IF AI6_#3.AI0 > 4000                 # raw integer, no scaling
IF $VAR.12 != 0
IF $SYS.ETH_LINK == 0
IF @Pwr-1.va >= 220

Operators: == != > >= < <= ANY . An unknown operator falls back to == . Values: ON / TRUE  → 1, OFF / FALSE  →

0, everything else is read as a decimal integer. The threshold must be a literal — it cannot be another tag.

Scheduling conditions

IF EVERY 30s                         # -> TIMER mode
IF DAILY AT 22:30                    # -> SCHEDULE, every day
IF WEEKLY MON,TUE,FRI AT 08:30       # -> SCHEDULE, selected days

EVERY  resolves to whole seconds; anything under 1 s is forced to 1 s with a UART warning, and the ceiling is 65535

s. Day names are SUN MON TUE WED THU FRI SAT , comma-separated with no spaces. DAILY / WEEKLY  match hour and

minute against the RTC and fire once per minute window.

AND / OR additional conditions

RULE pump_permit :
  IF  DI16_#2.DI0 == 1
  AND AI6_#3.AI0 > 500
  AND $VAR.5 == 0
  THEN DO16_#1.DO0 = 1
  ON LEVEL

The first extra clause fills cond2, the second fills cond3, further clauses are dropped. Evaluation is strictly left-

associative with no parentheses: ((c1 ⊕₁ c2) ⊕₂ c3) .

TRAP

AND  does not bind tighter than OR . In HRL, A OR B AND C  is (A OR B) AND C  — not the usual Boolean

precedence. Stage mixed expressions through $VAR .

# Written intent:  A OR (B AND C)
# HRL evaluates:  (A OR B) AND C
  IF  DI16_#2.DI0 == 1
  OR  DI16_#2.DI1 == 1
  AND AI6_#3.AI0 > 500

# Correct approach - stage it through $VAR
RULE or_stage :
  IF  DI16_#2.DI0 == 1
  OR  DI16_#2.DI1 == 1
  THEN $VAR.30 = 1
  ON LEVEL

RULE or_stage_r :
  IF  DI16_#2.DI0 == 0
  AND DI16_#2.DI1 == 0
  THEN $VAR.30 = 0
  ON LEVEL

RULE combined :
  IF  $VAR.30 == 1
  AND AI6_#3.AI0 > 500
  THEN DO16_#1.DO0 = 1
  ON LEVEL

If any condition's tag fails to read, the whole rule is invalid for that scan: the remembered last condition result and

any running dwell timer are both cleared, and an e:cond  entry appears in the trigger history. So a module going

offline wipes the rule's edge memory, and coming back online produces a spurious rising edge.

FOR — dwell time

FOR 250ms        ->    250 ms
FOR 5s           ->   5000 ms
FOR 0.5s         ->    500 ms
FOR 1.25s        ->   1250 ms
FOR 30           ->  30000 ms   (bare number = seconds)
FOR 1m           ->  60000 ms
FOR 2m           ->  65535 ms   CLAMPED + UART warning

While the dwell is unmet the engine treats the remembered last condition result as false, so FOR  combined with ON

RISE  correctly produces a rising edge on the scan the dwell completes. This behaviour is also what makes the self-

oscillating flasher work (§09).

TRAP

FOR  applies to the 

COMBINED RESULT OF ALL THREE CONDITIONS

, not to one of them. In a multi-condition rule any condition moving restarts the dwell timer. Debounce logic must

be isolated into its own single-condition rule writing to a $VAR .

The FOR  duration has a 

MAXIMUM OF 65535 MS (65.5 S)

. FOR 2m  is clamped to 65535 ms.

THEN actions

THEN DO16_#1.DO0 = 1
THEN DO16_#1.DO0 = ON                # identical
THEN AO4_#7.AO1  = 2048
THEN $VAR.3      = 0
THEN DO16_#1.DO0 = TOGGLE
THEN DO16_#1.DO0 = PULSE 500ms       # ON now, auto OFF after 500 ms
THEN ALARM "bilge high" HIGH
THEN MQTT plant/line1/state
THEN TRIGGER other_rule IMMEDIATE
THEN NTP_SYNC
THEN LOG

Syntax Behaviour

<tag> = <value> Write a literal

<tag> = <operand> <op> <literal> Arithmetic / bitwise expression

<tag> = TOGGLE Read current value, write its inverse

<tag> = PULSE [<dur>] Write 1 now, auto-write 0 after the given duration (default 3000 ms)

ALARM "<name>" <sev> Raise an alarm: INFO WARN HIGH CRITICAL

MQTT <topic> Publish to MQTT

TRIGGER <rule> [IMMEDIATE] Fire another rule

NTP_SYNC Trigger an NTP time sync

LOG  / NOOP Log only

Expressions

Operators + - * / % & | ^ . When the literal is 0, /  returns the operand and %  returns 0 — no divide-by-zero fault.

# Slot 1 (THEN) - operand may be ANY tag
THEN $VAR.10 = AI6_#3.AI0 / 10       # scale down
THEN $VAR.11 = $VAR.11 + 1           # counter
THEN AO4_#7.AO0 = AI6_#3.AI1 * 2     # analog follower
THEN $VAR.12 = DI16_#2.DI0 + 0       # COPY idiom
THEN $VAR.13 = $VAR.13 & 254         # clear bit 0
THEN $VAR.14 = $SYS.RTC_MIN % 15     # quarter-hour phase

# Slots 2 and 3 (AND) - operand must be $VAR.n
  AND $VAR.20 = $VAR.21 + 100        # OK
  AND $VAR.20 = AI6_#3.AI0 + 100     # NOT OK - operand ignored, writes 100

Slot Operand may be Stored in

Action 1 ( THEN ) Any tag — DI/DO/AI/AO/ $VAR / $SYS / @ext Full tag name

Actions 2 / 3 ( AND ) $VAR.n  only $VAR  index only

TRAP

The engine supports a plain copy internally, but HRL has 

NO WAY TO WRITE IT

. THEN DO = $VAR.5  is read as a literal and yields 

0

— with no error whatsoever.

THEN DO16_#1.DO0 = $VAR.5            # WRONG - writes 0, silently
THEN DO16_#1.DO0 = $VAR.5 + 0        # correct
THEN DO16_#1.DO0 = $VAR.5 * 1        # also correct

ON trigger modes
Defaults to LEVEL  when omitted; an unknown string also falls back to LEVEL . EVERY  / DAILY  / WEEKLY  set the mode

implicitly from the IF  line, so no ON  is needed.

Mode Semantics

LEVEL Condition true → actions execute every tick

RISE Once, on the tick the condition goes false→true

FALL Once, on the tick the condition goes true→false

CHANGE Once on any change of the condition

BOOT Once at startup; skipped entirely during normal ticks

TRIGGER Only woken by another rule; an IF , if present, acts as a gate

OFFDELAY  / TOF Off-delay, equivalent to IEC 61131-3 TOF

OFFDELAY (TOF)

RULE fan_runon :
  IF DI16_#2.DI5 == 1
  THEN DO16_#1.DO3 = 1
  ON OFFDELAY FOR 30s

Condition true → write action_value  immediately (or 1 if it is 0) and cancel any pending off-countdown. Condition

false → start the countdown; 0 is written only when it expires. FOR 0  degenerates to plain LEVEL.

TRAP

The FOR  in ON OFFDELAY FOR ...  is the 

OFF-DELAY

and occupies the rule's only duration field. The same rule therefore 

CANNOT ALSO USE IF ... FOR  FOR INPUT DEBOUNCE

. Chain two rules to get debounce followed by off-delay.

Rule chaining

RULE detect :
  IF AI6_#3.AI0 > 3800
  THEN TRIGGER shutdown IMMEDIATE
  ON RISE

RULE shutdown :
  THEN DO16_#1.DO0 = 0
  AND  DO16_#1.DO1 = 0
  AND  ALARM "overpressure trip" CRITICAL
  ON TRIGGER

TRIGGER  runs the target next tick (event semantics); TRIGGER … IMMEDIATE  runs it this tick (call semantics). Chain

guards: one fire per rule per tick, 64 fires per tick globally, IMMEDIATE recursion depth 8, worklist 16.

Boot rules

RULE init_state :
  IF $SYS.UPTIME >= 0
  THEN $VAR.0 = 0
  AND  $VAR.8 = 0
  AND  DO16_#1.DO0 = 0
  ON BOOT

COOLDOWN and attributes
COOLDOWN  is the minimum re-trigger interval, maximum 65535 ms. The first fire is exempt. When cooldown blocks a

rule the actions do not run, but the remembered last condition result is still updated — so the RISE edge is

swallowed.

THEN <tag> = PULSE <dur>  uses the same duration field as a standalone COOLDOWN  line. Do not use both in one rule.

Attribute Syntax Notes

GROUP GROUP 3  / GROUP Pumps 1–32; a string is hashed ( h = h*31 + c ) % 32 + 1. Groups can be enabled/disabled as a unit.

PRIORITY PRIORITY 10 1 = highest, 99 = lowest, default 50. Does not affect execution order.

DISABLED DISABLED enable = 0 ; the rule is kept but never runs.

TRAP — PRIORITY DOES NOT ORDER EXECUTION

PRIORITY  is persisted and exported, but the engine evaluates rules top to bottom on every scan and never

consults the field.

EXECUTION ORDER = ARRAY INDEX ORDER = THE ORDER RULES ARE WRITTEN IN THE HRL

FILE.

To control sequencing, control the file order.

Tag namespaces
Prefix Namespace Example R W

(none) Local IO modules DI16_#2.DI0 ✔ ✔

$SYS. System variables $SYS.UPTIME ✔ read-only

$VAR. User variables 0–63 $VAR.12 ✔ ✔

@ External devices @Pwr-1.va ✔ ✔

DI16_#2.DI0        slave 2, coil point 0,  digital input
DO16_#1.DO15       slave 1, coil point 15, digital output
DO6TR_#5.DO3       slave 5, coil point 3,  relay output
AI6_#3.AI0         slave 3, register 0,    analog input
AO4_#7.AO1         slave 7, register 1,    analog output
$VAR.42            user variable, 32-bit signed, retained
$SYS.RTC_HOUR      system variable, read-only
@Pwr-1.va          external device via ext_device

How local IO tags are actually parsed

Local tag parsing is far looser than it appears:

Find the first # ; the decimal digits after it are the slave id

Scan the whole string: if the substring AI  or AO  occurs → register; otherwise coil

The trailing run of digits at the end of the string is the point index

The slave id must resolve to a module that is online, otherwise it is not a local tag

In other words the model prefix plays no part in matching: DO16_#1.DO7  and XX_#1.7  resolve identically. Still write

the canonical form, because the Web UI and the Modbus map key off full names. Coil points are limited to 0–15.

Module ID Class

DI16 0x471C 16-channel digital input

DO16 0x471D 16-channel digital output

AI6 0x471F 6-channel analog input

DO6TR 0x4720 6-channel relay output

Complete $SYS list

Name Contents

UPTIME Seconds since boot

RTC_YEAR  … RTC_SEC RTC year / month / day / hour / minute / second

RTC_WEEK 0 = Sunday … 6 = Saturday

CPU_LOAD CPU load percentage

RULE_ENABLED Rule engine master switch 0/1

Name Contents

RULE_COUNT Number of active rules

MQTT_CONNECTED MQTT connected 0/1

ETH_LINK Ethernet link 0/1

SLAVE_COUNT Online IO module count

WQ_FAIL Cumulative writes dropped by a full write queue

Names are case-sensitive and must be uppercase. $VAR  is a 32-bit signed integer held in non-volatile memory and

preserved across reboots — add an ON BOOT  clearing rule if that is not what you want.

Scan cycle and execution model
The engine repeats the three stages below at a fixed period. The default is 20 ms (50 Hz), adjustable from 5 to 100

ms.

1  INPUT      every local input point is sampled into a frozen
              snapshot; the output image starts as a copy of it

2  EVALUATE   every rule is evaluated in the order it appears in
              the file, then any rule woken by TRIGGER IMMEDIATE

3  UPDATE     only the points actually written during this scan
              are sent out to the field bus

The critical read/write asymmetry

Tag class Value read this tick When a write takes effect

Local IO (DI/DO/AI/AO) the snapshot frozen at tick start flushed at tick end; later rules in the same tick cannot see it

$VAR  / $SYS  / @ext the live value immediately; later rules in the same tick see it

This is deliberate: freezing local IO eliminates same-tick chatter, while live $VAR  lets a state machine advance fully

within one scan.

TRAP

NEVER USE A DO POINT AS AN INTERMEDIATE BIT.

A rule cannot read a DO value written earlier in the same tick, so the state machine lags a full scan and its

behaviour shifts with rule order. Route all intermediate logic through $VAR .

Multiple rules writing the same DO

The later writer wins — both write into the same output image, and the update stage emits only the final value. This

makes layered override (write a default first, let later conditions overwrite it) the single most useful HRL technique: it

sidesteps the 3-condition limit, the lack of parentheses and the AND/OR precedence problem all at once.

# Layer 1 - default OFF
RULE out_default :
  IF $VAR.21 == 0
  THEN DO16_#1.DO1 = 0
  ON LEVEL

# Layer 2 - steady ON
RULE out_steady :
  IF  $VAR.21 == 1
  AND $VAR.1 == 0
  THEN DO16_#1.DO1 = 1
  ON LEVEL

# Layer 3 - flashing overrides steady
RULE out_flash :
  IF $VAR.1 == 1
  THEN DO16_#1.DO1 = $VAR.0 + 0
  ON LEVEL

# Layer 4 - lamp test overrides everything (must be LAST in file)
RULE out_lamptest :
  IF DI16_#2.DI1 == 1
  THEN DO16_#1.DO1 = 1
  ON LEVEL

The write queue

Writes beyond 96 physical points in one scan are dropped; the cumulative count is exposed as $SYS.WQ_FAIL  so a

rule can monitor itself:

RULE wq_watch :
  IF $SYS.WQ_FAIL > 0
  THEN ALARM "write queue full" HIGH
  AND  MQTT diag/wqfail
  ON RISE
  COOLDOWN 60s

Function block cookbook
Every pattern below uses only FOR , ON RISE/FALL  and $VAR , and is verified against V2.77 semantics.

IEC 61131-3 HRL equivalent

TON IF … FOR t  + ON RISE , output to a $VAR

TOF Native ON OFFDELAY FOR t

R_TRIG / F_TRIG Native ON RISE  / ON FALL

SR / RS Two ON RISE  rules writing the same $VAR

CTU / CTD $VAR = $VAR + 1  under ON RISE

MOVE dst = src + 0  (action slot 1)

BLINK Two interlocking FOR  + ON RISE  rules

SEL / MUX Layered override (§08)

IEC 61131-3 HRL equivalent

HYSTERESIS Web UI / JSON API only — no HRL syntax

TON — on-delay

RULE ton_ls1 :
  IF DI16_#2.DI2 == 1 FOR 5s
  THEN $VAR.21 = 1
  ON RISE

RULE ton_ls1_r :
  IF DI16_#2.DI2 == 0
  THEN $VAR.21 = 0
  ON RISE

Always isolate this into a single-condition rule. A FOR  on a multi-condition rule gets restarted by any other condition

moving. Ceiling is 65.5 s.

SR — set/reset latch

# Reset-dominant: reset rule is written AFTER set rule
RULE trip_set :
  IF AI6_#3.AI0 > 3800
  THEN $VAR.9 = 1
  ON RISE

RULE trip_reset :
  IF DI16_#2.DI0 == 1
  THEN $VAR.9 = 0
  ON RISE

Set-dominant vs reset-dominant is decided by file order — whichever is written later wins. The example above is

reset-dominant.

Flasher (scan-period independent)

RULE blink_on :
  IF $VAR.0 == 0 FOR 500ms
  THEN $VAR.0 = 1
  ON RISE

RULE blink_off :
  IF $VAR.0 == 1 FOR 500ms
  THEN $VAR.0 = 0
  ON RISE

# consume it
RULE lamp_flash :
  IF $VAR.1 == 1
  THEN DO16_#1.DO1 = $VAR.0 + 0
  ON LEVEL

It oscillates because the engine treats the remembered last condition result as false while the dwell is unmet,

guaranteeing a rising edge the moment it completes. 500 ms ON / 500 ms OFF = 1 Hz. More flexible than IF EVERY

1s THEN $VAR.0 = TOGGLE , which is capped at 0.5 Hz by the 1-second EVERY  floor.

Counter

RULE cnt_up :
  IF DI16_#2.DI6 == 1
  THEN $VAR.30 = $VAR.30 + 1
  ON RISE

RULE cnt_done :
  IF $VAR.30 >= 100
  THEN DO16_#1.DO5 = 1
  AND  $VAR.30 = 0
  ON RISE

RULE cnt_clear :
  IF $SYS.UPTIME >= 0
  THEN $VAR.30 = 0
  ON BOOT

Start / stop latch with trip

RULE motor_start :
  IF  DI16_#2.DI0 == 1
  AND $VAR.40 == 0
  THEN $VAR.40 = 1
  ON RISE

RULE motor_stop :
  IF DI16_#2.DI1 == 1
  THEN $VAR.40 = 0
  ON RISE

RULE motor_trip :
  IF DI16_#2.DI2 == 1
  THEN $VAR.40 = 0
  ON LEVEL

RULE motor_out :
  IF $VAR.40 >= 0
  THEN DO16_#1.DO0 = $VAR.40 + 0
  ON LEVEL

motor_trip  uses ON LEVEL  rather than ON RISE : while the trip condition holds it must keep the output down, not

clear it once on an edge. It sits after motor_stop  in the file, so trip dominates start.

Duty / standby pump alternation

RULE lead_swap :
  IF $VAR.50 == 0
  THEN $VAR.50 = 1
  ON RISE

RULE lead_swap_b :
  IF $VAR.50 == 1
  THEN $VAR.50 = 0
  ON RISE

RULE pump_a :
  IF  $VAR.41 == 1
  AND $VAR.50 == 0
  THEN DO16_#1.DO2 = 1
  ON LEVEL

RULE pump_b :
  IF  $VAR.41 == 1
  AND $VAR.50 == 1
  THEN DO16_#1.DO3 = 1
  ON LEVEL

Communications watchdog

RULE bus_degraded :
  IF $SYS.SLAVE_COUNT < 4
  THEN ALARM "IO module offline" CRITICAL
  AND  DO16_#1.DO7 = 1
  ON RISE
  COOLDOWN 30s

RULE bus_ok :
  IF $SYS.SLAVE_COUNT >= 4
  THEN DO16_#1.DO7 = 0
  ON RISE

Scheduling

RULE night_dim :
  IF DAILY AT 22:00
  THEN AO4_#7.AO0 = 400

RULE day_bright :
  IF DAILY AT 06:00
  THEN AO4_#7.AO0 = 2000

Worked example: bilge high-level alarm panel
A seven-point marine bilge alarm annunciator replacing a 20-point PLC plus seven H3Y-4 timing relays. Behaviour:

unacknowledged points flash at 1 Hz with the buzzer sounding; the acknowledge button silences the buzzer and

makes the lamp steady; the lamp clears when the level recovers; a new point after acknowledgement re-sounds the

buzzer. 54 rules total.

# ---- I/O map --------------------------------------------------
# DI16_#2.DI0     BS   buzzer stop / acknowledge
# DI16_#2.DI1     LT   lamp test pushbutton (wired direct to DI)
# DI16_#2.DI2~8   LS1..LS7 field level switches
# DO16_#1.DO0     BZ   buzzer      (DO16 PNP sourcing output, direct drive)
# DO16_#1.DO1~7   lamps R1..R7 (DO16 PNP sourcing output, no dimming)
# DO6TR_#5.DO0~5  CRT repeat 1..6  (dry contacts)
# DO6TR_#6.DO0    CRT repeat 7
# DO6TR_#6.DO1~2  local panel repeats (former T4 / T5 6-10 contacts)
#
# ---- $VAR map -------------------------------------------------
# $VAR.0          FLASH  1 Hz source
# $VAR.1~7        UNACK1..7
# $VAR.8          BUZ latch
# $VAR.21~27      DLY1..7  debounced alarm state (replaces H3Y-4)

The two core rules per alarm point

RULE p1_ton :
  IF DI16_#2.DI2 == 1 FOR 5s
  THEN $VAR.21 = 1
  AND  $VAR.1  = 1
  AND  $VAR.8  = 1
  ON RISE

RULE p1_rst :
  IF DI16_#2.DI2 == 0
  THEN $VAR.21 = 0
  AND  $VAR.1  = 0
  ON RISE

p1_ton  does three things at once: latch the debounced DLY state, flag it unacknowledged, and set the buzzer latch.

Three actions fill the rule's action slots exactly. The FOR 5s  replaces the original H3Y-4's 0–30 s adjustable delay,

which existed to stop ship roll producing false alarms.

The acknowledge button

RULE ack_a :
  IF DI16_#2.DI0 == 1
  THEN $VAR.8 = 0
  AND  $VAR.1 = 0
  AND  $VAR.2 = 0
  ON RISE

RULE ack_b :
  IF DI16_#2.DI0 == 1
  THEN $VAR.3 = 0
  AND  $VAR.4 = 0
  AND  $VAR.5 = 0
  ON RISE

RULE ack_c :
  IF DI16_#2.DI0 == 1
  THEN $VAR.6 = 0
  AND  $VAR.7 = 0
  ON RISE

One rising edge must clear eight flags but a rule has only three action slots, so it is split into three rules sharing the

same condition. Each remembers its own condition state, so all three fire on the same edge with no TRIGGER

chaining. They must be written before every p*_ton , otherwise a new alarm arriving in the same tick would be

wrongly silenced.

Lamp output

DO16 provides PNP sourcing outputs, so the indicator lamps and buzzer hang directly off the output points — no

interposing relays and no dimming. The original drawing's DM:5K potentiometer, dimmer module and RY1

interposing relay are all omitted here, and the lamp-test pushbutton wires straight into a DI.

The lamp's three states (off / steady / flashing) plus the lamp-test override use the layered pattern from §08 — see

the example there. The four layers appear in file order, with the lamp-test layer last in the whole file.

Repeat dry contacts

RULE p1_rep :
  IF $VAR.21 >= 0
  THEN DO6TR_#5.DO0 = $VAR.21 + 0
  ON LEVEL

TRAP

The repeat contacts must follow DLY , 

NOT THE LAMP OUTPUT

. On the original drawing they are physical timer contacts, unrelated to the PLC's acknowledge logic — after the

operator acknowledges, the engine control room must still see an active alarm. Wiring them to the lamp would

send the remote panel a stream of false repeat alarms once flashing starts.

Rule ordering in the file

ON BOOT  initialisation

Flasher ×2

Acknowledge ×3

Debounce + reset ×14

Lamp output, three layers ×21

Repeats ×9

Buzzer ×1

Lamp test ×3 (must be last)

Diagnostic tools and boot behaviour

Trigger history

Opened from the History button at the bottom of the Rules page. The last 64 entries are kept, with time, rule and

result.

• 

• 

• 

• 

• 

11

16 / 20



Result Meaning

skip Condition false, not triggered

trig Executed normally

dry Simulation mode, nothing written

e:cond A condition tag failed to read

e:act An action failed to write

e:chain Chain limit exceeded, or the TRIGGER  target name was not found

e:wq Outputs were dropped this scan because the queue was full

Debug mode

Entered from the Debug button on the Rules page. It breaks a rule's evaluation into seven stages and shows the

value the engine actually read and the verdict at each one — the only tool that answers "why did this rule not fire".

Stage What it checks

1 READ_TAG Whether the condition tag reads, and what value

2 Condition 1 Result of the comparison against the threshold

3 Condition 2 combine The result after the extra conditions are combined with AND / OR

4 Dwell Whether the FOR  duration has been met

5 Edge detection Whether the ON  mode says to fire this scan

6 Cooldown Whether the COOLDOWN  interval is still blocking

7 Execute The target and value actually written

The scope can be all rules, the currently selected rule, or one group; the monitor interval is 0.5 / 1 / 2 / 5 s. There is

also a Force button that runs the actions once, and Inject, which writes a value that makes the condition true —

injection is impossible when the source is a physical input, so change the real input or use a virtual module instead.

Watch and cross reference

The watch table pins any tag and shows its live value. Cross reference lists which rules read and which rules write

that tag — it answers "what is actually moving this value". The anomaly filter shows only rules with a failed tag read

or an execution error; a condition simply being false is not an anomaly.

Hold freezes the engine with outputs kept as they are (this is not disable-and-clear); it auto-resumes after 10

minutes idle. Step releases exactly one scan. Force overrides up to 16 points, beats every other writer, and is

cleared by a reboot; rules keep evaluating throughout.

Simulation mode (dry run)

A switch in Engine Settings. It blocks all physical outputs, MQTT and alarms, but lets $VAR  writes through so state

machines still advance. Note that those $VAR  values are real and are not rolled back when simulation ends.

Self-monitoring with rules

These system variables can go straight into a rule, turning diagnostics into an active alarm:

Variable What it watches

$SYS.WQ_FAIL Cumulative outputs dropped by a full queue

$SYS.SLAVE_COUNT Online IO module count — alarm when one drops

$SYS.ETH_LINK Ethernet link

$SYS.MQTT_CONNECTED MQTT connection

$SYS.CPU_LOAD CPU load percentage

RULE wq_watch :
  IF $SYS.WQ_FAIL > 0
  THEN ALARM "write queue full" HIGH
  AND  MQTT diag/wqfail
  ON RISE
  COOLDOWN 60s

Boot strategy

Set from the boot-rule dropdown in Engine Settings:

Option Behaviour

BOOT first (skip safety) Skip the safety policy and run ON BOOT  rules directly (default)

Safety then BOOT (zero first) The safety policy zeroes every DO, waits 1 s, then BOOT rules run

Safety only (no BOOT) BOOT rules are not run

ON BOOT  rules are skipped entirely during normal scans. They run at startup, when the engine master switch goes

off→on, on a replace-mode HRL import, or when the manual fire-BOOT button in Engine Settings is pressed.

Trap index
# Trap Remedy

1 PRIORITY  does not order execution Control file order

2 Local IO reads a frozen snapshot; same-tick writes are

invisible

Route intermediates through $VAR

3 THEN dst = $VAR.5  silently writes 0 Write = $VAR.5 + 0

4 Action slots 2/3 accept only $VAR.n  operands Put AI/$SYS copies in action slot 1

5 FOR  applies to the combined condition Isolate debounce into a single-condition rule

6 AND  does not bind tighter than OR Left-associative; use layered override instead

7 FOR  / COOLDOWN  cap at 65535 ms Use $SYS.UPTIME  or a TIMER counter

8 Beyond 3 conditions / 3 actions is silently dropped Split into several rules

12

# Trap Remedy

9 ON OFFDELAY FOR  consumes the dwell field Chain two rules

10 PULSE <dur>  and COOLDOWN  share a field Use one or the other per rule

11 Unrecognised lines are silently ignored Verify rule count and contents after import

12 Duplicate rule names resolve to the first only Unique names, within 23 bytes, ASCII recommended

13 A failed condition read wipes edge memory Expect one spurious rising edge when a module returns

14 Cooldown still updates the remembered condition The RISE edge is swallowed; use FOR  to damp chatter

instead

15 Imports over ~3.5 KB are silently truncated Strip comments and import in append chunks

Export / import asymmetry

HRL text exported from the device is not guaranteed to re-import faithfully:

Exported form Problem

EVERY 30s The parser only accepts IF EVERY ; a standalone line is ignored → the TIMER rule degrades to LEVEL

DAILY AT 22:30 Same; the SCHEDULE is lost

THEN dst = operand The COPY form re-parses as literal 0

AND ALARM "name" Severity is not emitted for action slots 2/3

Hysteresis Not emitted at all

BOTTOM LINE

Treat the HRL file as the project's 

SINGLE SOURCE OF TRUTH

and keep it under version control. Do not rely on an export-then-reimport round trip; export is for human

inspection only.

Quick reference card

RULE name :                          rule header, 23 chars max
  IF tag op literal [FOR dur]        condition 1   (op: == != > >= < <=)
  AND tag op literal                 condition 2   (OR also allowed)
  AND tag op literal                 condition 3   (left-associative)
  THEN dst = value                   action 1  (any-tag expression operand)
  AND  dst = value                   action 2  ($VAR-only operand)
  AND  dst = value                   action 3  ($VAR-only operand)
  ON LEVEL|RISE|FALL|CHANGE|BOOT|TRIGGER|OFFDELAY
  COOLDOWN dur                       min re-trigger interval
  GROUP n     PRIORITY n     DISABLED

IF EVERY dur | IF DAILY AT HH:MM | IF WEEKLY DAY,DAY AT HH:MM

value:   123 | ON | OFF | TRUE | FALSE | TOGGLE | PULSE [dur]
         operand + - * / % & | ^ literal
action:  tag = value | ALARM "x" INFO|WARN|HIGH|CRITICAL
         MQTT topic | TRIGGER rule [IMMEDIATE] | NTP_SYNC | LOG | NOOP
dur:     250ms | 5s | 0.5s | 1m | 1h        max 65535 ms
tag:     DI16_#n.DIm  DO16_#n.DOm  DO6TR_#n.DOm  AI6_#n.AIm
         $VAR.0-63    $SYS.NAME    @Device.field

Print this section as a field reference card. For full semantics return to the relevant chapter.

